

**meta regress postestimation** — Postestimation tools for meta regress

[Postestimation commands](#)

[predict](#)

[margins](#)

[Remarks and examples](#)

[Methods and formulas](#)

[Reference](#)

[Also see](#)

## Postestimation commands

The following postestimation command is of special interest after `meta regress`:

| Command                       | Description  |
|-------------------------------|--------------|
| <code>estat bubbleplot</code> | bubble plots |

The following postestimation commands are available after `meta regress`:

| Command                      | Description   |
|------------------------------|---|
| <code>contrast</code>        | contrasts and ANOVA-style joint tests of estimates  |
| <code>estat summarize</code> | summary statistics for the estimation sample  |
| <code>estat vce</code>       | variance–covariance matrix of the estimators (VCE)  |
| <code>estimates</code>       | cataloging estimation results   |
| <code>lincom</code>          | point estimates, standard errors, testing, and inference for linear combinations of coefficients    |
| <code>margins</code>         | marginal means, predictive margins, marginal effects, and average marginal effects                  |
| <code>marginsplot</code>     | graph the results from margins (profile plots, interaction plots, etc.)                             |
| <code>nlcom</code>           | point estimates, standard errors, testing, and inference for nonlinear combinations of coefficients |
| <code>predict</code>         | predictions, residuals, and other diagnostic measures   |
| <code>predictnl</code>       | point estimates, standard errors, testing, and inference for generalized predictions                |
| <code>pwcompare</code>       | pairwise comparisons of estimates   |
| <code>test</code>            | Wald tests of simple and composite linear hypotheses  |
| <code>testnl</code>          | Wald tests of nonlinear hypotheses  |

# predict

## Description for predict

`predict` creates a new variable containing predictions such as linear predictions, residuals, leverage, and standard errors. After random-effects meta-regression, you can also obtain estimates of random effects and their standard errors.

## Menu for predict

Statistics > Postestimation

## Syntax for predict

*Syntax for obtaining BLUPs of random effects and their standard errors after RE meta-regression*

```
predict [type] newvar [if] [in], reffects [se(newvar)]
```

*Syntax for obtaining other predictions*

```
predict [type] newvar [if] [in] [, statistic fixedonly se(sespec)]
```

| <i>statistic</i>            | Description  |
|-----------------------------|--|
| Main                        |  |
| <code>xb</code>             | linear prediction; the default   |
| <code>stdp</code>           | standard error of the linear prediction                                      |
| <code>fitted</code>         | fitted values, fixed-portion linear prediction plus predicted random effects |
| <code>residuals</code>      | residuals, response minus fitted values                                      |
| <code>leverage   hat</code> | leverage (diagonal elements of hat matrix)                                   |

Unstarred statistics are available both in and out of sample; type `predict ... if e(sample) ...` if wanted only for the estimation sample.

## Options for predict

Main

`xb`, the default, calculates the linear prediction  $\mathbf{x}_j\hat{\beta}$ . For the random-effects meta-regression, this corresponds to the fixed portion of the linear predictor based on the estimated regression coefficients. That is, this is equivalent to fixing all random effects in the model to their theoretical mean value of 0.

`stdp` calculates the standard error of the linear prediction.

`reffects` calculates best linear unbiased predictions (BLUPs) of the random effects.

`fitted` calculates the fitted values. With fixed-effects meta-regression or with random-effects meta-regression when option `fixedonly` is also specified, this option is equivalent to `xb`. For random-effects meta-regression without `fixedonly`, it calculates  $\mathbf{x}_j\hat{\beta} + u_j$ , which is equal to the fixed portion of the linear prediction plus predicted random effects.

`residuals` calculates the residuals, which are equal to the responses minus the fitted values. With fixed-effects meta-regression or with random-effects meta-regression when option `fixedonly` is also specified, it calculates  $\hat{\theta}_j - \mathbf{x}_j\hat{\beta}$ . The former are known as marginal residuals in the context of the random-effects model. For random-effects meta-regression without `fixedonly`, this option calculates  $\hat{\theta}_j - (\mathbf{x}_j\hat{\beta} + u_j)$ , which are known as conditional residuals.

`leverage` or `hat` calculates the diagonal elements of the projection (“hat”) matrix.

`fixedonly` specifies that all random effects be set to zero, which is equivalent to using only the fixed portion of the model, when computing results for random-effects models. This option may be specified only with statistics `fitted`, `residuals`, or `leverage`.

`se(newvar[, marginal])` calculates the standard error of the corresponding predicted values. This option may be specified only with statistics `reffects`, `fitted`, and `residuals`.

Suboption `marginal` is allowed only with random-effects meta-regression and requires option `fixedonly`. It computes marginal standard errors, when you type

```
. predict ..., statistic se(newvar, marginal) fixedonly
```

instead of the standard errors conditional on zero random effects, which are computed when you type

```
. predict ..., statistic se(newvar) fixedonly
```

`marginal` is not allowed in combination with `reffects`.

## margins

### Description for margins

`margins` estimates margins of response for linear predictions.

### Menu for margins

Statistics > Postestimation

### Syntax for margins

```
margins [marginlist] [, options]
```

```
margins [marginlist] , predict(statistic ...) [options]
```

| <i>statistic</i>                         | Description                                   |
|--|---|
| <code>xb</code>                          | linear prediction; the default                |
| <code>fitted</code>                      | fitted values; implies <code>fixedonly</code> |
| <code>stdp</code>                        | not allowed with <code>margins</code>         |
| <code>reffects</code>                    | not allowed with <code>margins</code>         |
| <code>residuals</code>                   | not allowed with <code>margins</code>         |
| <code>leverage</code>   <code>hat</code> | not allowed with <code>margins</code>         |

Statistics not allowed with `margins` are functions of stochastic quantities other than  $e(b)$ .

For the full syntax, see [R] [margins](#).

## Remarks and examples

We demonstrate some of the postestimation features, including `estat bubbleplot`, `margins`, and `predict` after `meta regress`.

### ► Example 1: Bubble plot

Consider the declared [BCG dataset](#) of clinical trials that studied the efficacy of a Bacillus Calmette-Guérin (BCG) vaccine in the prevention of tuberculosis (TB) (Colditz et al. 1994). In [example 1 of \[META\] meta regress](#), we used `meta regress` to fit a simple meta-regression to these data with the continuous moderator `latitude_c` to explore heterogeneity.

```
. use https://www.stata-press.com/data/r16/bcgset
(Efficacy of BCG vaccine against tuberculosis; set with -meta esize-)
. meta regress latitude_c

Effect-size label:  Log Risk-Ratio
Effect size:       _meta_es
Std. Err.:        _meta_se


Random-effects meta-regression                Number of obs =      13
Method: REML                                Residual heterogeneity:
                                           tau2 = .07635
                                           I2 (%) = 68.39
                                           H2 = 3.16
                                           R-squared (%) = 75.63
Wald chi2(1) = 16.36
Prob > chi2 = 0.0001
```

| _meta_es   | Coef.     | Std. Err. | z     | P> z  | [95% Conf. Interval] |
|------------|-----------|-----------|-------|-------|----------------------|
| latitude_c | -.0291017 | .0071953  | -4.04 | 0.000 | -.0432043 -.0149991  |
| _cons      | -.7223204 | .1076535  | -6.71 | 0.000 | -.9333174 -.5113234  |

Test of residual homogeneity:  $Q_{res} = \chi^2(11) = 30.73$  Prob >  $Q_{res} = 0.0012$

Whenever there is one continuous moderator in a meta-regression, a so-called bubble plot is commonly used to explore the relationship between the effect size and that moderator. Let's use `estat bubbleplot` to produce the bubble plot after the fitted meta-regression.

```
. estat bubbleplot
```



A bubble plot is a scatterplot of the observed effect sizes against the moderator overlaid with the predicted regression and confidence-intervals lines. Each study is represented by a circle (bubble) with the size (area) proportional to the study precision,  $1/\hat{\sigma}_j^2$ . The larger the size of the bubble, the more precise the study. The coordinates of the center of each circle show the observed value of the effect size on the  $y$  axis and that of the moderator (`latitude_c` in our example) on the  $x$  axis. The solid line shows the predicted values (predicted log risk-ratios in our example). The predicted 95% confidence intervals are also plotted.

From the plot, the log risk-ratio for the BCG vaccine declines as the distance from the equator increases. There appear to be a couple of outlying studies (see points in the bottom left and middle top sections of the plot), but their bubbles are very small, which suggests that their log risk-ratios estimates had small weights, relative to other studies, in the meta-regression. Outlying studies with large bubbles may be a source of concern because of the large differences in their effect sizes compared with those from the other studies and because of the large impact they have on the regression results. ↵

## ▷ Example 2: Marginal effects

Continuing with [example 1](#), we found that the log risk-ratio for the BCG decreases as the distance from the equator increases. For example, from the bubble plot, a trial conducted relatively close to the equator, say, in Thailand (with a latitude of 15 or a centered latitude of  $-18.5$ ), would have a predicted log risk-ratio of about  $-0.2$ . A trial conducted in, say, Nepal (with a latitude of 28 or a centered latitude of  $-5.5$ ), would have a predicted log risk-ratio of about  $-0.7$ . And a trial conducted in, say, Ukraine (with a latitude of 50 or a centered latitude of  $16.5$ ), would have a predicted log risk-ratio of about  $-1$ .

Instead of relying on the graph, we can obtain more precise estimates of the predicted log risk-ratios at different latitude values by using the `margins` command as follows:

```
. margins, at(latitude_c = (-18.5 -5.5 16.5))
Adjusted predictions          Number of obs      =          13
Expression   : Fitted values; fixed portion (xb), predict(fitted fixedonly)
1._at        : latitude_c      =          -18.5
2._at        : latitude_c      =           -5.5
3._at        : latitude_c      =           16.5
```

|     | Delta-method |           |       |       | [95% Conf. Interval] |           |
|-----|--------------|-----------|-------|-------|----------------------|-----------|
|     | Margin       | Std. Err. | z     | P> z  |                      |           |
| _at |              |           |       |       |                      |           |
| 1   | -.1839386    | .1586092  | -1.16 | 0.246 | -.4948069            | .1269297  |
| 2   | -.562261     | .1091839  | -5.15 | 0.000 | -.7762574            | -.3482645 |
| 3   | -1.202499    | .1714274  | -7.01 | 0.000 | -1.53849             | -.8665072 |

The list of numbers specified in the `at()` option are the values of the latitudes centered around the latitude mean ( $\approx 33.5$ ).

Note that results produced by `margins` are on the log scale and need to be exponentiated to make interpretations on the natural (risk) scale. For instance, from the output, the risk ratio for regions with `latitude_c = 16.5` is  $\exp(-1.202499) = 0.3$ , which means that the vaccine is expected to reduce the risk of TB by 70% for regions with that latitude.

### Example 3: Predicted random effects


In [example 1](#), we noticed a couple of outlying studies. Let's explore this further by looking at predicted random effects from our random-effects meta-regression.

We first use `predict` with options `reffects` and `se()` to predict the random-effects and estimate their standard errors.

```
. predict double u, reffects se(se_u)
```

Then, we generate a new variable, `ustandard`, as the ratio of the predicted random effects to their standard errors and use the `qnorm` command (see [\[R\] Diagnostic plots](#)) to construct the normal quantile plot.

```
. generate double ustandard = u/se_u
. label variable ustandard "Standardized predicted random effects"
. qnorm ustandard, mlabel(trial)
```



The plot suggests that trial 7, labeled “Vandiviere et al., 1973” in our data, is an outlier. From the data, the log risk-ratio estimate for this trial is  $-1.62$  with the corresponding risk-ratio estimate of about 0.2. This means that, in that trial, the vaccine reduced the risk of TB by roughly 80% even though this trial was conducted relatively close to the equator (in Haiti, with `latitude=19`). In fact, this trial reported the largest risk reduction (smallest log-risk-ratio value) in the meta-analysis. Compare this with trial 11 (“Comstock et al., 1974”), which was conducted in Puerto Rico and has a similar latitude (`latitude=18`) but whose estimated risk reduction was much more moderate, about 29% (with the risk-ratio estimate of  $\exp(-0.34) = 0.71$ ). More investigation is needed to explain the extreme value reported by trial 7. Thus, in this example, you may consider reporting the results of meta-analyses with and without this trial.

4

## Methods and formulas

Methods and formulas are presented under the following headings:

[Random-effects meta-regression](#)  
[Fixed-effects meta-regression](#)

The following formulas are used by `predict`. The notation is based on *Methods and formulas of [META] meta regress*.

## Random-effects meta-regression

The fixed-portion of the linear prediction (option `xb`) is  $\mathbf{x}_j \widehat{\boldsymbol{\beta}}$ . The estimated standard error of the fixed-portion of the linear prediction (option `stdp`) for study  $j$  is

$$\widehat{\text{SE}}(\mathbf{x}_j \widehat{\boldsymbol{\beta}}) = \sqrt{\mathbf{x}_j (\mathbf{X}' \mathbf{W}^* \mathbf{X})^{-1} \mathbf{x}_j'}$$

The BLUP of the  $j$ th random effect (option `reffects`) is

$$\widehat{u}_j = \lambda_j (\widehat{\theta}_j - \mathbf{x}_j \widehat{\boldsymbol{\beta}})$$

where

$$\lambda_j = \frac{\widehat{\tau}^2}{\widehat{\tau}^2 + \widehat{\sigma}_j^2}$$

is the empirical Bayes shrinkage factor for the  $j$ th study. When the `se()` option is also specified, the estimated standard error of  $\widehat{u}_j$  is

$$\widehat{\text{SE}}(\widehat{u}_j) = \lambda_j \sqrt{\widehat{\sigma}_j^2 + \widehat{\tau}^2 - \mathbf{x}_j (\mathbf{X}' \mathbf{W}^* \mathbf{X})^{-1} \mathbf{x}_j'}$$

The fitted value (option `fitted`) is

$$\widetilde{\theta}_j = \mathbf{x}_j \widehat{\boldsymbol{\beta}} + \widehat{u}_j$$

When the `se()` option is also specified, the estimated standard error of  $\widetilde{\theta}_j$  is

$$\widehat{\text{SE}}(\widetilde{\theta}_j) = \sqrt{\lambda_j^2 (\widehat{\sigma}_j^2 + \widehat{\tau}^2) + (1 - \lambda_j^2) \mathbf{x}_j (\mathbf{X}' \mathbf{W}^* \mathbf{X})^{-1} \mathbf{x}_j'}$$

The residual (option `residuals`) is

$$e_j = \widehat{\theta}_j - \widetilde{\theta}_j$$

When option `se()` is also specified, the estimated standard error of  $e_j$  is

$$\widehat{\text{SE}}(e_j) = \sqrt{(1 + \lambda_j^2) (\widehat{\sigma}_j^2 + \widehat{\tau}^2 - \mathbf{x}_j (\mathbf{X}' \mathbf{W}^* \mathbf{X})^{-1} \mathbf{x}_j')}$$

The leverage (option `hat`) are the diagonal elements of the hat matrix  $\mathbf{X} (\mathbf{X}' \mathbf{W}^* \mathbf{X})^{-1} \mathbf{X}' \mathbf{W}^*$ :

$$h_j^* = \frac{1}{\widehat{\tau}^2 + \widehat{\sigma}_j^2} \mathbf{x}_j (\mathbf{X}' \mathbf{W}^* \mathbf{X})^{-1} \mathbf{x}_j'$$

When the `fixedonly` option is specified, the formulas for the fitted values and residuals (including their standard errors) and leverage are adjusted by replacing the value of  $\hat{u}_j$  with 0, in which case,  $\hat{\tau}^2 = 0$ ,  $\lambda_j = 0$ , and  $\mathbf{W}^*$  is replaced with  $\mathbf{W} = \text{diag}(1/\hat{\sigma}_1^2, \dots, 1/\hat{\sigma}_K^2)$ . In this case, the standard errors are computed conditionally on zero random effects.

If `se()`'s option `marginal` is specified, then marginal standard errors are computed. This is equivalent to computing  $\widehat{\text{SE}}(\tilde{\theta}_j)$  and  $\widehat{\text{SE}}(e_j)$  with  $\lambda_j = 0$  but keeping  $\hat{\tau}^2$  and  $\mathbf{W}^*$  unchanged.

## Fixed-effects meta-regression

The linear prediction (option `xb`) is  $\mathbf{x}_j \hat{\beta}$ . The estimated standard error of the linear prediction (option `stdp`) for study  $j$  is

$$\widehat{\text{SE}}(\mathbf{x}_j \hat{\beta}) = \sqrt{\mathbf{x}_j (\mathbf{X}' \mathbf{W} \mathbf{X})^{-1} \mathbf{x}_j'}$$

The fitted value (option `fitted`) is the same as the linear prediction:

$$\tilde{\theta}_j = \mathbf{x}_j \hat{\beta}$$

The residual (option `residuals`) is

$$e_j = \hat{\theta}_j - \tilde{\theta}_j$$

When option `se()` is also specified, the estimated standard error of  $e_j$  is

$$\widehat{\text{SE}}(e_j) = \sqrt{(\hat{\sigma}_j^2 - \mathbf{x}_j (\mathbf{X}' \mathbf{W} \mathbf{X})^{-1} \mathbf{x}_j')}$$

The leverage (option `hat`) are the diagonal elements of the hat matrix  $\mathbf{X} (\mathbf{X}' \mathbf{W} \mathbf{X})^{-1} \mathbf{X}' \mathbf{W}$ :

$$h_j = \frac{1}{\hat{\sigma}_j^2} \mathbf{x}_j (\mathbf{X}' \mathbf{W} \mathbf{X})^{-1} \mathbf{x}_j'$$

For the multiplicative fixed-effects meta-regression, in the above formulas, replace  $\mathbf{W}$  with  $\mathbf{W}^\phi$  and  $\hat{\sigma}_j^2$  with  $\hat{\phi} \hat{\sigma}_j^2$ , where  $\hat{\phi}$  is defined in *Fixed-effects meta-regression* in [META] **meta regress**.

## Reference

- Colditz, G. A., T. F. Brewer, C. S. Berkey, M. E. Wilson, E. Burdick, H. V. Fineberg, and F. Mosteller. 1994. Efficacy of BCG vaccine in the prevention of tuberculosis: Meta-analysis of the published literature. *Journal of the American Medical Association* 271: 698–702.



## Also see

[META] [meta regress](#) — Meta-analysis regression

[META] [meta data](#) — Declare meta-analysis data

[META] [meta](#) — Introduction to meta

[META] [Glossary](#)

[META] [Intro](#) — Introduction to meta-analysis

[U] [20 Estimation and postestimation commands](#)